ALV Object Model – Hierarchical Sequential List – The Basics

Applies to:

SAP NetWeaver 2004 and SAP NetWeaver 2004s

Summary

In this tutorial, you will learn the basic steps to create a hierarchical-sequential list using the ALV Object Model. For more examples, see any program, which start with SALV*.

Author(s): Rich Heilman

Company: Yorktowne Cabinetry, Inc.

Created on: 10 October 2006

Author Bio

Rich Heilman is an ABAP/J2EE Software Engineer/Analyst for Yorktowne Cabinetry, Inc. based in Red Lion, Pennsylvania, USA. He has a total of nine years experience in the IT industry. He has spent the past five years studying ABAP and Java.

Table of Contents

Main Class - CL_SALV_HIERSEQ_TABLE

The main class used to create the hierarchical-sequential list is CL_SALV_HIERSEQ_TABLE. First, create a reference variable for this class. Create two internal tables with types SCARR and SFLIGHT. Fill the internal tables with data from the database. Create the ALV object using the FACTORY method. This method expects that you pass the two internal tables with data and another internal table, which describes the binding of the two internal tables. Create an internal table called IBINDING, and add a record to this internal table as seen below. Create the object of the ALV list using the FACTORY method of the class CL_SALV_HIERSEQ_TABLE, and then call the method DISPLAY to display the list.

```
report  zalv_omhsl_1.

data: gr_table  type ref to cl_salv_hierseq_table.
data: iscarr type table of scarr.
data: isflight type table of sflight.
data: ibinding type salv_t_hierseq_binding.
data: xbinding type salv_s_hierseq_binding.

select * into table iscarr from scarr.
select * into table isflight from sflight.

xbinding-master = 'CARRID'.
xbinding-slave  = 'CARRID'.
append  xbinding to ibinding.

cl_salv_hierseq_table=>factory(
    exporting
      t_binding_level1_level2 = ibinding
    importing
      r_hierseq = gr_table
    changing
      t_table_level1 = iscarr
      t_table_level2 = isflight ).

gr_table->display( ).
```


The output should be similarly to the figure below.



Cli	Air	Airline	Curre	Airline URL			
Cli	Air	Flig Date	Airfare	Curre	Plane Type	Capacity	Occupied
000	AA	American Airlines	USD	http://www.aa.com			
000	AA	17 03/30/2005	422.94	USD	747-400	385	375
000	AA	17 04/27/2005	422.94	USD	747-400	385	367
000	AA	17 05/25/2005	422.94	USD	747-400	385	367
000	AA	17 06/22/2005	422.94	USD	747-400	385	371
000	AA	17 07/20/2005	422.94	USD	747-400	385	373
000	AA	17 08/17/2005	422.94	USD	747-400	385	373
000	AA	17 09/14/2005	422.94	USD	747-400	385	372
000	AA	17 10/12/2005	422.94	USD	747-400	385	374
000	AA	17 11/09/2005	422.94	USD	747-400	385	371
000	AA	17 12/07/2005	422.94	USD	747-400	385	170
000	AA	17 01/04/2006	422.94	USD	747-400	385	76
000	AA	17 02/01/2006	422.94	USD	747-400	385	29
000	AA	17 03/01/2006	422.94	USD	747-400	385	14
000	AA	17 03/29/2006	422.94	USD	747-400	385	42
000	AA	17 04/26/2006	422.94	USD	747-400	385	22
000	AA	64 04/01/2005	422.94	USD	A310-300	280	271
000	AA	64 04/29/2005	422.94	USD	A310-300	280	265
000	AA	64 05/27/2005	422.94	USD	A310-300	280	271
000	AA	64 06/24/2005	422.94	USD	A310-300	280	271
000	AA	64 07/22/2005	422.94	USD	A310-300	280	268
000	AA	64 08/19/2005	422.94	USD	A310-300	280	271
000	AA	64 09/16/2005	422.94	USD	A310-300	280	270

Functions – CL_SALV_FUNCTIONS

Next, add functions to the application toolbar. For this, use the CL_SALV_FUNCTIONS class. Create the object reference variable and receive the object using the GET_FUNCTIONS method of the GR_TABLE object. Call the method SET_ALL to force the ALV grid to show all standard functions.

```
report  zalv_omhsl_1.

data: gr_table  type ref to cl_salv_hierseq_table.
data: gr_functions type ref to cl_salv_functions.

data: iscarr type table of scarr.
data: isflight type table of sflight.

data: ibinding type salv_t_hierseq_binding.
data: xbinding type salv_s_hierseq_binding.

select * into table iscarr from scarr.
select * into table isflight from sflight.

xbinding-master = 'CARRID'.
xbinding-slave  = 'CARRID'.
append  xbinding to ibinding.

cl_salv_hierseq_table=>factory(
    exporting
      t_binding_level1_level2 = ibinding
    importing
      r_hierseq = gr_table
    changing
      t_table_level1 = iscarr
      t_table_level2 = isflight ).

gr_functions = gr_table->get_functions( ).
gr_functions->set_all( abap_true ).

gr_table->display( ).
```


The output should be similarly to the figure below.



Columns – CL_SALV_COLUMNS_HIERSEQ

Using the class CL_SALV_COLUMNS_HIERSEQ, we can manipulate the columns of the table, for example, providing the expand/collapse functionality. First, we need to add a field to the internal table called ISCARR. This field will be used to toggle the expand/collapse functionality. Create a TYPE statement, which describes the structure as you see below. Next, create reference variables for GR_COLUMNS and GR_COLUMN. Modify the internal table declaration to use the new type, which you defined earlier. Receive the object GR_COLUMNS using the method GET_COLUMNS of the object GR_TABLE. Next, use the method SET_EXPAND_COLUMN to set the specific column used for toggling the expand/collapse functionality.

```
report  zalv_omhsl_1.

types: begin of t_scarr.
        include structure scarr.
types: expcol type c.
types: end of t_scarr.

data: gr_table  type ref to cl_salv_hierseq_table.
data: gr_functions type ref to cl_salv_functions.

data: gr_columns type ref to cl_salv_columns_hierseq.
data: gr_column  type ref to cl_salv_column_hierseq.

data: iscarr type table of t_scarr.
data: isflight type table of sflight.

data: ibinding type salv_t_hierseq_binding.
data: xbinding type salv_s_hierseq_binding.

select * into table iscarr from scarr.
select * into table isflight from sflight.

xbinding-master = 'CARRID'.
xbinding-slave  = 'CARRID'.
append  xbinding to ibinding.

cl_salv_hierseq_table=>factory(
    exporting t_binding_level1_level2 = ibinding
    importing r_hierseq = gr_table
    changing
      t_table_level1 = iscarr
      t_table_level2 = isflight ).

gr_functions = gr_table->get_functions( ).
gr_functions->set_all( abap_true ).

gr_columns = gr_table->get_columns( level = 1 ).
gr_columns->set_expand_column( 'EXPCOL' ).


gr_table->display( ).
```


The output should be similarly to the figure below.



Levels – CL_SALV_HIERSEQ_LEVEL

The next requirement is that the user wants the expand/collapse functionality but wants all items to be expanded by default. To do this, we will use the class CL_SALV_HIERSEQ_LEVEL. Create the reference variable GR_LEVEL. Next receive the object from the GET_LEVEL method of the GR_TABLE object. Call the SET_ITEMS_EXPANDED method.

```
report  zalv_omhsl_1.

types: begin of t_scarr.
        include structure scarr.
types: expcol type c.
types: end of t_scarr.

data: gr_table  type ref to cl_salv_hierseq_table.
data: gr_functions type ref to cl_salv_functions.
data: gr_columns type ref to cl_salv_columns_hierseq.
data: gr_column  type ref to cl_salv_column_hierseq.

data: gr_level type ref to cl_salv_hierseq_level.

data: iscarr type table of t_scarr.
data: isflight type table of sflight.
data: ibinding type salv_t_hierseq_binding.
data: xbinding type salv_s_hierseq_binding.

select * into table iscarr from scarr.
select * into table isflight from sflight.

xbinding-master = 'CARRID'.
xbinding-slave  = 'CARRID'.
append  xbinding to ibinding.

cl_salv_hierseq_table=>factory(
    exporting t_binding_level1_level2 = ibinding
    importing r_hierseq = gr_table
    changing
      t_table_level1 = iscarr
      t_table_level2 = isflight ).

gr_functions = gr_table->get_functions( ).
gr_functions->set_all( abap_true ).

gr_columns = gr_table->get_columns( level = 1 ).
gr_columns->set_expand_column( 'EXPCOL' ).

gr_level = gr_table->get_level( 1 ).
gr_level->set_items_expanded( ).

gr_table->display( ).
```

The output should be similarly to the figure below.



ALV OM - Hierarchical-Sequential List Demo

Cli	Air	Airline / Flig	Date	Curre	Airline URL / Airfare	Plane Type	Capacity	Occupied
000	AA	American Airlines		USD	http://www.aa.com			
000	AA	17	03/30/2005		422.94 USD	747-400	385	375
000	AA	17	04/27/2005		422.94 USD	747-400	385	367
000	AA	17	05/25/2005		422.94 USD	747-400	385	367
000	AA	17	06/22/2005		422.94 USD	747-400	385	371
000	AA	17	07/20/2005		422.94 USD	747-400	385	373
000	AA	17	08/17/2005		422.94 USD	747-400	385	373
000	AA	17	09/14/2005		422.94 USD	747-400	385	372
000	AA	17	10/12/2005		422.94 USD	747-400	385	374
000	AA	17	11/09/2005		422.94 USD	747-400	385	371
000	AA	17	12/07/2005		422.94 USD	747-400	385	170
000	AA	17	01/04/2006		422.94 USD	747-400	385	76
000	AA	17	02/01/2006		422.94 USD	747-400	385	29
000	AA	17	03/01/2006		422.94 USD	747-400	385	14
000	AA	17	03/29/2006		422.94 USD	747-400	385	42
000	AA	17	04/26/2006		422.94 USD	747-400	385	22
000	AA	64	04/01/2005		422.94 USD	A310-300	280	271
000	AA	64	04/29/2005		422.94 USD	A310-300	280	265

Sorts – CL_SALV_SORTS

The next requirement is that the output should show the detail data sorted by Flight Date in descending order. For this, we will utilize the class CL_SALV_SORTS. Receive the object from the GET_SORTS method of the object GR_TABLE, specifying the LEVEL. Then call the method ADD_SORTS of the object GR_SORTS, specifying the column name and the sequence.

```
report  zalv_omhsl_1.

types: begin of t_scarr.
        include structure scarr.
types: expcol type c.
types: end of t_scarr.

data: gr_table  type ref to cl_salv_hierseq_table.
data: gr_functions type ref to cl_salv_functions.

data: gr_columns type ref to cl_salv_columns_hierseq.
data: gr_column  type ref to cl_salv_column_hierseq.

data: gr_level type ref to cl_salv_hierseq_level.

data: gr_sorts type ref to cl_salv_sorts.

data: iscarr type table of t_scarr.
data: isflight type table of sflight.

data: ibinding type salv_t_hierseq_binding.
data: xbinding type salv_s_hierseq_binding.

select * into table iscarr from scarr.
select * into table isflight from sflight.

xbinding-master = 'CARRID'.
xbinding-slave  = 'CARRID'.
append  xbinding to ibinding.

cl_salv_hierseq_table=>factory(
    exporting
      t_binding_level1_level2 = ibinding
    importing
      r_hierseq = gr_table
    changing
      t_table_level1 = iscarr
      t_table_level2 = isflight ).

gr_functions = gr_table->get_functions( ).
gr_functions->set_all( abap_true ).

gr_columns = gr_table->get_columns( level = 1 ).
gr_columns->set_expand_column( 'EXPCOL' ).

gr_level = gr_table->get_level( 1 ).
gr_level->set_items_expanded( ).

gr_sorts = gr_table->get_Sorts( level = '2' ).
```



```
gr_sorts->add_sort( columnname = 'FLDATE'
                    sequence = if_salv_c_sort=>sort_down ).

gr_table->display( ).
```

The output should be similarly to the figure below.



Cli	Air	Airline / Flig	Date	Curre	Airline URL / Airfare	Curre	Plane Type	Capacity	Occupied
000	AA	American Airlines		USD	http://www.aa.com				
000	AA	64	04/28/2006		422.94	USD	A310-300	280	7
000	AA	17	04/26/2006		422.94	USD	747-400	385	22
000	AA	64	03/31/2006		422.94	USD	A310-300	280	8
000	AA	17	03/29/2006		422.94	USD	747-400	385	42
000	AA	64	03/03/2006		422.94	USD	A310-300	280	39
000	AA	17	03/01/2006		422.94	USD	747-400	385	14
000	AA	64	02/03/2006		422.94	USD	A310-300	280	2
000	AA	17	02/01/2006		422.94	USD	747-400	385	29
000	AA	64	01/06/2006		422.94	USD	A310-300	280	61
000	AA	17	01/04/2006		422.94	USD	747-400	385	76
000	AA	64	12/09/2005		422.94	USD	A310-300	280	17
000	AA	17	12/07/2005		422.94	USD	747-400	385	170
000	AA	64	11/11/2005		422.94	USD	A310-300	280	270
000	AA	17	11/09/2005		422.94	USD	747-400	385	371
000	AA	64	10/14/2005		422.94	USD	A310-300	280	263
000	AA	17	10/12/2005		422.94	USD	747-400	385	374
000	AA	64	09/16/2005		422.94	USD	A310-300	280	270
000	AA	17	09/14/2005		422.94	USD	747-400	385	372
000	AA	64	08/19/2005		422.94	USD	A310-300	280	271
000	AA	17	08/17/2005		422.94	USD	747-400	385	373
000	AA	64	07/22/2005		422.94	USD	A310-300	280	268
000	AA	17	07/20/2005		422.94	USD	747-400	385	373
000	AA	64	06/24/2005		422.94	USD	A310-300	280	271
000	AA	17	06/22/2005		422.94	USD	747-400	385	371
000	AA	64	05/27/2005		422.94	USD	A310-300	280	271
000	AA	17	05/25/2005		422.94	USD	747-400	385	367
000	AA	64	04/29/2005		422.94	USD	A310-300	280	265
000	AA	17	04/27/2005		422.94	USD	747-400	385	367
000	AA	64	04/01/2005		422.94	USD	A310-300	280	271
000	AA	17	03/30/2005		422.94	USD	747-400	385	375

Related Content

- Help - ALV Object Model
- Utilizing the New ALV Object Model
- SDN ABAP Forum

Disclaimer and Liability Notice

This document may discuss sample coding or other information that does not include SAP official interfaces and therefore is not supported by SAP. Changes made based on this information are not supported and can be overwritten during an upgrade.

SAP will not be held liable for any damages caused by using or misusing the information, code or methods suggested in this document, and anyone using these methods does so at his/her own risk.

SAP offers no guarantees and assumes no responsibility or liability of any type with respect to the content of this technical article or code sample, including any liability resulting from incompatibility between the content within this document and the materials and services offered by SAP. You agree that you will not hold, or seek to hold, SAP responsible or liable with respect to the content of this document.